

ÉQUATIONS DIFFÉRENTIELLES ORDINAIRES DU PREMIER ORDRE : SOLUTIONS DES EXERCICES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Exercice M1

Énoncé

Déterminer la solution générale des EDO linéaires du premier ordre suivantes.

1) $y'(t) + 2y(t) = 4t^2$

2) $y'(t) + y(t) = 2e^t$

3) $y'(t) - 2y(t) = 2t^3 + t$

4) $y'(t) - 3y(t) = 2e^{3t}$

5) $y'(t) - y(t) = t + e^t$

6) $y'(t) - 2y(t) = 4\cos(2t)$

7) $y'(t) + y(t) = 2te^{-t}$

Solution

Ces questions sont tirées de l'ouvrage de J-P Lepoutre et P. Pilibossian *Analyse II. Exercices corrigés avec rappels de cours* (Masson, collection *Mathématiques pour les sciences économiques*, 3^e édition). Pour de telles EDO linéaires du premier ordre à coefficient constant, la réussite de **dsolve** est de 100%, que le second membre soit polynômial, exponentiel, trigonométrique ou mixte.

1)

```
> eqd1:=diff(y(t),t)+2*y(t)=4*t^2;  
soleqd1:=dsolve(eqd1,y(t));
```

$$eqd1 := \frac{d}{dt} y(t) + 2y(t) = 4t^2$$

$$soleqd1 := y(t) = 1 - 2t + 2t^2 + e^{-2t} _C1$$

2)

```
> eqd2:=diff(y(t),t)+y(t)=2*exp(t);  
soleqd2:=dsolve(eqd2,y(t));
```

$$eqd2 := \frac{d}{dt} y(t) + y(t) = 2e^t$$

$$soleqd2 := y(t) = e^t + e^{-t} _C1$$

3)

```
> eqd3:=diff(y(t),t)-2*y(t)=2*t^3+t;  
soleqd3:=dsolve(eqd3,y(t));
```

$$eqd3 := \frac{d}{dt} y(t) - 2y(t) = 2t^3 + t$$

$$\text{soleqd3} := y(t) = -1 - 2t - \frac{3}{2}t^2 - t^3 + e^{2t} _CI$$

4)

```
> eqd4:=diff(y(t),t)-3*y(t)=2*exp(3*t);
soleqd4:=dsolve(eqd4,y(t));
```

$$\text{eqd4} := \frac{d}{dt} y(t) - 3y(t) = 2e^{3t}$$

$$\text{soleqd4} := y(t) = (2t + _CI) e^{3t}$$

5)

```
> eqd5:=diff(y(t),t)-y(t)=t+exp(t);
soleqd5:=dsolve(eqd5,y(t));
```

$$\text{eqd5} := \frac{d}{dt} y(t) - y(t) = t + e^t$$

$$\text{soleqd5} := y(t) = -t - 1 + e^t t + e^t _CI$$

6)

```
> eqd6:=3*diff(y(t),t)-2*y(t)=4*cos(2*t);
soleqd6:=dsolve(eqd6,y(t));
```

$$\text{eqd6} := 3 \left(\frac{d}{dt} y(t) \right) - 2y(t) = 4 \cos(2t)$$

$$\text{soleqd6} := y(t) = -\frac{1}{5} \cos(2t) + \frac{3}{5} \sin(2t) + e^{\frac{2}{3}t} _CI$$

7)

```
> eqd7:=diff(y(t),t)+y(t)=2*t*exp(-t);
soleqd7:=dsolve(eqd7,y(t));
```

$$\text{eqd7} := \frac{d}{dt} y(t) + y(t) = 2e^{-t}t$$

$$\text{soleqd7} := y(t) = (t^2 + _CI) e^{-t}$$

Exercice M2

Enoncé

Déterminer la solution générale des EDO linéaires du premier ordre suivantes.

1) $y'(t) - 2ty(t) = (1 - 2t)e^t$

2) $y'(t) - 2ty(t) = \sin(t)e^{2t}$

3) $y'(t) - \cos(t)y(t) = \cos(t)$

4) $y'(t) - \cos(t)y(t) = \sin(2t)$

5) $y'(t) \sin(t) - \cos(t)y(t) = 3t^2 \sin(t)^2$

6) $(1 - t^2)y'(t) - 2ty(t) = t^2$

7) $(1 + t^2)y'(t) + ty(t) - 2t = 0$

8) $t(-1 + t)y'(t) - (2t - 1)y(t) + t^2 = 0$

Solution

Comme dans l'exercice précédent, les questions sont extraites de Lepoutre et Pilibossian (op. cité). Il s'agit d'EDO linéaires à coefficients variables que la commande **dsolve** résout sans difficulté.

1) Pour $y'(t) - 2ty(t) = (1 - 2t)e^t$, on a :

```
> eqd1:=diff(y(t),t)-2*t*y(t)=(1-2*t)*exp(t);soleqd1:=dsolve  
(eqd1,y(t));
```

$$eqd1 := \frac{d}{dt} y(t) - 2ty(t) = (1 - 2t)e^t$$

$$soleqd1 := y(t) = (e^{-t(-1+t)} + _CI) e^{t^2}$$

2) $y'(t) - 2ty(t) = \sin(t)e^{2t}$:

```
> eqd2:=diff(y(t),t)-2*t*y(t)=sin(t)*exp(t^2);soleqd2:=dsolve  
(eqd2,y(t));
```

$$eqd2 := \frac{d}{dt} y(t) - 2ty(t) = \sin(t)e^{t^2}$$

$$soleqd2 := y(t) = -e^{t^2} \cos(t) + e^{t^2} _CI$$

3) $y'(t) - \cos(t)y(t) = \cos(t)$:

```
> eqd3:=diff(y(t),t)-cos(t)*y(t)=cos(t);soleqd3:=dsolve(eqd3,y  
(t));
```

$$eqd3 := \frac{d}{dt} y(t) - \cos(t)y(t) = \cos(t)$$

$$soleqd3 := y(t) = -1 + e^{\sin(t)} _CI$$

4) $y'(t) - \cos(t)y(t) = \sin(2t)$:

```
> eqd4:=diff(y(t),t)-cos(t)*y(t)=sin(2*t);soleqd4:=dsolve(eqd4,  
y(t));
```

$$eqd4 := \frac{d}{dt} y(t) - \cos(t)y(t) = \sin(2t)$$

$$soleqd4 := y(t) = -2 \sin(t) - 2 + e^{\sin(t)} _CI$$

5) $y'(t) \sin(t) - \cos(t)y(t) = 3t^2 \sin(t)^2$:

```
> eqd5:=diff(y(t),t)*sin(t)-cos(t)*y(t)=3*t^2*sin(t)^2;  
soleqd5:=dsolve(eqd5,y(t));
```

$$eqd5 := \left(\frac{d}{dt} y(t) \right) \sin(t) - \cos(t)y(t) = 3t^2 \sin(t)^2$$

$$soleqd5 := y(t) = \sin(t)t^3 + \sin(t) _CI$$

6) $(1 - t^2)y'(t) - 2ty(t) = t^2$:

```
> eqd6:=(1-t^2)*diff(y(t),t)-2*t*y(t)=t^2;soleqd6:=dsolve(eqd6,  
y(t));
```

$$eqd6 := (1 - t^2) \left(\frac{d}{dt} y(t) \right) - 2ty(t) = t^2$$

$$soleqd6 := y(t) = \frac{-\frac{1}{3}t^3 + _CI}{(-1+t)(t+1)}$$

7) $(1 + t^2)y'(t) + ty(t) - 2t = 0$:

```
> eqd7:=(1+t^2)*diff(y(t),t)+t*y(t)-2*t=0;soleqd7:=dsolve(eqd7,
```

$y(t)$);

$$eqd7 := (1 + t^2) \left(\frac{d}{dt} y(t) \right) + t y(t) - 2t = 0$$

$$soleqd7 := y(t) = 2 + \frac{-CI}{\sqrt{1 + t^2}}$$

8) $t(-1 + t)y'(t) - (2t - 1)y(t) + t^2 = 0$:

```
> eqd8:=t*(t-1)*diff(y(t),t)-(2*t-1)*y(t)+t^2=0;soleqd8:=dsolve  
(eqd8,y(t));
```

$$eqd8 := t(-1 + t) \left(\frac{d}{dt} y(t) \right) - (2t - 1)y(t) + t^2 = 0$$

$$soleqd8 := y(t) = t(-1 + t) _CI + t$$

Exercice M3

Enoncé

Résoudre les problèmes de Cauchy suivants. Donner à chaque fois une représentation de la trajectoire-solution.

1) $y'(t) + 2y(t) = 4t^2$ avec $y(0) = 3$.

2) $y'(t) + y(t) = 2e^t$ avec $y(0) = 0$.

3) $y'(t) - 2y(t) = 2t^3 + t$ avec $y(0) = 1$.

4) $y'(t) - 3y(t) = 2e^{3t}$ avec $y(0) = 4e^3$.

5) $y'(t) - y(t) = t + e^t$ avec $y(0) = -1$.

6) $y'(t) - 2y(t) = 4 \cos(2t)$ avec $y(0) = \frac{17}{20}$.

7) $y'(t) + y(t) = 2te^{-t}$ avec $y(0) = 1$.

Solution

Il suffit d'utiliser `dsolve` avec un premier argument de la forme **{equation différentielle, condition initiale}**. Après récupération de l'expression de la solution, on trace la trajectoire solution avec `plot`. L'option `numpoints` permet le plus souvent d'affiner la courbe.


1)

```
> eqd1:=diff(y(t),t)+2*y(t)=4*t^2;  
soleqd1:=dsolve({eqd1,y(0)=3},y(t));  
sol1:=subs(soleqd1,y(t));  
plot(sol1,t=0..10,numpoints=200);
```

$$eqd1 := \frac{d}{dt} y(t) + 2y(t) = 4t^2$$

$$soleqd1 := y(t) = 1 - 2t + 2t^2 + 2e^{-2t}$$

$$sol1 := 1 - 2t + 2t^2 + 2e^{-2t}$$




2)

```
> eqd2:=diff(y(t),t)+y(t)=2*exp(t);
soleqd2:=dsolve({eqd2,y(0)=0},y(t));
sol2:=subs(soleqd2,y(t));
plot(sol2,t=0..5,numpoints=200);
```

$$eqd2 := \frac{d}{dt} y(t) + y(t) = 2 e^t$$

$$soleqd2 := y(t) = e^t - e^{-t}$$

$$sol2 := e^t - e^{-t}$$




3)

```
> eqd3:=diff(y(t),t)-2*y(t)=2*t^3+t;
soleqd3:=dsolve({eqd3,y(0)=1},y(t));
sol3:=subs(soleqd3,y(t));
plot(sol3,t=0..5,numpoints=200);
```

$$eqd3 := \frac{d}{dt} y(t) - 2 y(t) = 2 t^3 + t$$

$$soleqd3 := y(t) = -1 - 2 t - \frac{3}{2} t^2 - t^3 + 2 e^{2t}$$

$$sol3 := -1 - 2 t - \frac{3}{2} t^2 - t^3 + 2 e^{2t}$$




4)

```
> eqd4:=diff(y(t),t)-3*y(t)=2*exp(3*t);
soleqd4:=dsolve({eqd4,y(0)=4*exp(1)^3},y(t));
sol4:=subs(soleqd4,y(t));
plot(sol4,t=0..2.5,numpoints=200);
```

$$eqd4 := \frac{d}{dt} y(t) - 3 y(t) = 2 e^{3t}$$

$$soleqd4 := y(t) = (2t + 4 (e)^3) e^{3t}$$

$$sol4 := (2t + 4 (e)^3) e^{3t}$$




5)

```
> eqd5:=diff(y(t),t)-y(t)=t+exp(t);
soleqd5:=dsolve({eqd5,y(0)=-1},y(t));
sol5:=subs(soleqd5,y(t));
plot(sol5,t=0..5,numpoints=200);
```

$$eqd5 := \frac{d}{dt} y(t) - y(t) = t + e^t$$

$$soleqd5 := y(t) = -t - 1 + e^t t$$

$$sol5 := -t - 1 + e^t t$$




6)

```
> eqd6:=3*diff(y(t),t)-2*y(t)=4*cos(2*t);
soleqd6:=dsolve({eqd6,y(0)=17/20},y(t));
sol6:=subs(soleqd6,y(t));
plot(sol6,t=0..4,numpoints=200);
```

$$eqd6 := 3 \left(\frac{d}{dt} y(t) \right) - 2 y(t) = 4 \cos(2 t)$$

$$soleqd6 := y(t) = -\frac{1}{5} \cos(2 t) + \frac{3}{5} \sin(2 t) + \frac{21}{20} e^{\frac{2}{3} t}$$

$$sol6 := -\frac{1}{5} \cos(2 t) + \frac{3}{5} \sin(2 t) + \frac{21}{20} e^{\frac{2}{3} t}$$




7)

```
> eqd7:=diff(y(t),t)+y(t)=2*t*exp(-t);
soleqd7:=dsolve({eqd7,y(0)=1},y(t));
sol7:=subs(soleqd7,y(t));
plot(sol7,t=0..10,numpoints=200);
```

$$eqd7 := \frac{d}{dt} y(t) + y(t) = 2 e^{-t} t$$

$$soleqd7 := y(t) = (t^2 + 1) e^{-t}$$

$$sol7 := (t^2 + 1) e^{-t}$$



Exercice M4

Enoncé

Résoudre les problèmes de Cauchy suivants. Donner à chaque fois une représentation de la trajectoire-solution.

- 1) $y'(t) - 2t y(t) = (1 - 2t) e^t$ avec $y(0) = 5$.
- 2) $y'(t) - 2t y(t) = \sin(t) e^{2t}$ avec $y(0) = 1$.
- 3) $y'(t) - \cos(t) y(t) = \cos(t)$ avec $y(0) = 0$.
- 4) $y'(t) - \cos(t) y(t) = \sin(2t)$ avec $y(0) = -1$.
- 5) $y'(t) \sin(t) - \cos(t) y(t) = 3t^2 \sin(t)^2$ avec $y(1) = 0$.
- 6) $(1 - t^2) y'(t) - 2t y(t) = t^2$ avec $y(0) = 1$.
- 7) $(1 + t^2) y'(t) + t y(t) - 2t = 0$ avec $y(1) = 3$.
- 8) $t(-1 + t)y'(t) - (2t - 1) y(t) + t^2 = 0$ avec $y(2) = 1$.

Solution

Même méthode que dans l'exercice précédent.

1)

```
> restart;
eqd1:=diff(y(t),t)-2*t*y(t)=(1-2*t)*exp(t);
```

```


soleqd1:=dsolve({eqd1,y(0)=5},y(t));
sol1:=subs(soleqd1,y(t));
plot(sol1,t=0..1,numpoints=200);

```

$$eqd1 := \frac{d}{dt} y(t) - 2 t y(t) = (1 - 2 t) e^t$$

$$soleqd1 := y(t) = (e^{-t(-1+t)} + 4) e^{t^2}$$

$$sol1 := (e^{-t(-1+t)} + 4) e^{t^2}$$



(2)

```


> restart;
eqd2:=diff(y(t),t)-2*t*y(t)=sin(t)*exp(t^2);
soleqd2:=dsolve({eqd2,y(0)=1 },y(t));
sol2:=subs(soleqd2,y(t));
plot(sol2,t=0..1.5,numpoints=200);

```

$$eqd2 := \frac{d}{dt} y(t) - 2 t y(t) = \sin(t) e^{t^2}$$

$$soleqd2 := y(t) = -e^{t^2} \cos(t) + 2 e^{t^2}$$

$$\text{sol2} := -e^{t^2} \cos(t) + 2 e^{t^2}$$




3)

```
> eqd3:=diff(y(t),t)-cos(t)*y(t)=cos(t);  
soleqd3:=dsolve({eqd3,y(0)=0},y(t));  
sol3:=subs(soleqd3,y(t));  
plot(sol3,t=0..25,numpoints=200);
```

$$\text{eqd3} := \frac{d}{dt} y(t) - \cos(t) y(t) = \cos(t)$$

$$\text{soleqd3} := y(t) = -1 + e^{\sin(t)}$$

$$\text{sol3} := -1 + e^{\sin(t)}$$




4)

```
> eqd4:=diff(y(t),t)-cos(t)*y(t)=sin(2*t);
soleqd4:=dsolve({eqd4,y(0)=-1},y(t));
sol4:=subs(soleqd4,y(t));
plot(sol4,t=0..15,numpoints=200);
```

$$eqd4 := \frac{d}{dt} y(t) - \cos(t) y(t) = \sin(2t)$$

$$soleqd4 := y(t) = -2 \sin(t) - 2 + e^{\sin(t)}$$

$$sol4 := -2 \sin(t) - 2 + e^{\sin(t)}$$




5)

```
> eqd5:=diff(y(t),t)*sin(t)-cos(t)*y(t)=3*t^2*sin(t)^2;
soleqd5:=dsolve({eqd5,y(1)=0},y(t));
sol5:=subs(soleqd5,y(t));
plot(sol5,t=0..50,numpoints=200);
```

$$eqd5 := \left(\frac{d}{dt} y(t) \right) \sin(t) - \cos(t) y(t) = 3 t^2 \sin(t)^2$$

$$soleqd5 := y(t) = \sin(t) t^3 - \sin(t)$$

$$sol5 := \sin(t) t^3 - \sin(t)$$




6)

```
> eqd6:=(1-t^2)*diff(y(t),t)-2*t*y(t)=t^2;
soleqd6:=dsolve({eqd6,y(0)=1},y(t));
sol6:=subs(soleqd6,y(t));
plot(sol6,t=-0.5..0.5,numpoints=200);
```

$$eqd6 := (1 - t^2) \left(\frac{d}{dt} y(t) \right) - 2ty(t) = t^2$$

$$soleqd6 := y(t) = \frac{-\frac{1}{3}t^3 - 1}{(t-1)(t+1)}$$

$$sol6 := \frac{-\frac{1}{3}t^3 - 1}{(t-1)(t+1)}$$




7)

```
> eqd7:=(1+t^2)*diff(y(t),t)+t*y(t)-2*t=0;
soleqd7:=dsolve({eqd7,y(1)=3},y(t));
sol7:=subs(soleqd7,y(t));
plot(sol7,t=0..5,numpoints=200);
```

$$eqd7 := (1 + t^2) \left(\frac{d}{dt} y(t) \right) + t y(t) - 2 t = 0$$

$$soleqd7 := y(t) = 2 + \frac{\sqrt{2}}{\sqrt{1+t^2}}$$

$$sol7 := 2 + \frac{\sqrt{2}}{\sqrt{1+t^2}}$$




8)

```
> eqd8:=t*(t-1)*diff(y(t),t)-(2*t-1)*y(t)+t^2=0;
soleqd8:=dsolve({eqd8,y(2)=1},y(t));
sol8:=subs(soleqd8,y(t));
plot(sol8,t=0..5,numpoints=200);
```

$$eqd8 := t(t-1) \left(\frac{d}{dt} y(t) \right) - (2t-1)y(t) + t^2 = 0$$

$$soleqd8 := y(t) = -\frac{1}{2} t(t-1) + t$$

$$sol8 := -\frac{1}{2} t(t-1) + t$$



Exercice E1 (dynamique du prix de marché)

Enoncé

On considère un marché très générique sur lequel l'offre agrégée et la demande agrégée dépendent respectivement positivement et négativement du prix courant. Dans la version linéarisée du modèle, on a : $Q_s(t) = a P(t) - b$ et $Q_d(t) = -c P(t) + d$, tous les paramètres étant positifs. La demande excédentaire à la date t est la différence entre l'offre et la demande courantes :

$Ed(t) = Q_d(t) - Q_s(t)$. L'ajustement du prix obéit à la règle suivante : un excès positif de demande accroît le prix; un excès de demande négatif - autrement dit un excès d'offre - fait baisser le prix; l'égalité de l'offre et de la demande laisse le prix au même niveau. La variation instantanée du prix s'interprétant comme la dérivée première de $P(t)$ par rapport à t , la traduction formelle la plus

simple de ce principe est linéaire : $\frac{d}{dt} P(t) = m Ed(t)$, le paramètre $m > 0$ mesurant l'intensité de la réaction du mécanisme d'ajustement du prix au déséquilibre courant.

1) Donner l'expression de la trajectoire-solution du prix de marché à partir de la condition initiale $P(0) = P_0$.

2) En microéconomie, la question est moins de savoir calculer la trajectoire suivie par le prix que de savoir s'il existe un prix d'équilibre - assurant par définition l'égalité des volumes offerts et demandés sur le marché - et si cet équilibre est stable - assurant qu'un écart du prix courant au prix d'équilibre se résorbe par le seul jeu des mécanismes de marché. Déterminer le prix d'équilibre puis

étudier sa nature. Tracer quelques trajectoires confirmant la convergence de toute trajectoire-solution vers le prix d'équilibre.

Solution

Avec très peu de moyens, Maple permet de répondre rapidement à ces questions. On commence par définir les fonctions d'offre, de demande et de demande excédentaire.

```
> restart;
Qs:=P->a*P-b;#fonction d'offre
Qd:=P->-c*P+d;#fonction de demande
Ed:=P->m*(Qd(P)-Qs(P));#fonction d'excès de demande
      Qs:=P→aP−b
      Qd:=P→−cP+d
      Ed:=P→m(Qd(P)−Qs(P))
```

(5.1)

1) La trajectoire du prix est donnée par **dsolve** appliquée à un problème de Cauchy.

```
> eqd:=diff(P(t),t)=Ed(P(t));#écriture de l'équation
d'évolution du prix
cint:=P(0)=P0;#condition initiale
dsolve({eqd,cint},P(t));#détermination de la trajectoire
solution générale
Psol:=subs(%,P(t));#récupération de l'expression de la
trajectoire
```

$$eqd := \frac{d}{dt} P(t) = m(-cP(t) + d - aP(t) + b)$$

$$cint := P(0) = P0$$

$$P(t) = e^{-m(c+a)t} \left(P0 - \frac{d+b}{c+a} \right) + \frac{d+b}{c+a}$$

$$Psol := e^{-m(c+a)t} \left(P0 - \frac{d+b}{c+a} \right) + \frac{d+b}{c+a}$$

(5.2)

2) La détermination du prix d'équilibre fait appel à la commande **solve**.

```
> Pe:=solve(Ed(P)=0,P);#détermination du prix d'équilibre
```

$$Pe := \frac{d+b}{c+a}$$

(5.3)

En invoquant **limit**, on obtient le comportement asymptotique de *Psol*. Il faut préalablement poser les hypothèses économiques sur les paramètres *a*, *b*, *c*, *d*, *m* du modèle, faute de quoi Maple manque d'informations pour répondre correctement et sans ambiguïté.

```
> assume(a>0,b>0,c>0,d>0,m>0);#écriture des hypothèses sur les
paramètres du modèle
limit(Psol,t=infinity);#recherche de la limite
```

$$\frac{d\sim + b\sim}{c\sim + a\sim}$$

(5.4)

Quel que soit le prix de départ, la trajectoire solution converge clairement vers le prix d'équilibre qui est par conséquent globalement stable comme le confirme la représentation graphique suivante.

```
> a,b,c,d,m:=0.1,-5,0.25,10,0.1;#attribution de valeurs aux
paramètres
```

```


Psol;
Psolve:=unapply(Psol,(P0,t));#Une trajectoire solution
dépend du prix initial et de la date t.
plot([seq(Psolve(Pe+i,t),i=-10..10,2.5)],t=0..100,P=0..30);
#représentation graphique de 9 trajectoires.

```

$a, b, c, d, m := 0.1, -5, 0.25, 10, 0.1$

$$e^{-0.035t} (P0 - 14.28571429) + 14.28571429$$

$$Psolve := (P0, t) \rightarrow e^{-0.035t} (P0 - 14.28571429) + 14.28571429$$



Exercice E2 (dynamique du prix de marché)

Enoncé

Reprendre les questions de l'exercice précédent en supposant à présent que la demande dépend négativement du prix courant mais aussi des anticipations parfaites de variations du prix. Autrement dit, on pose : $Qd(t) = -c P(t) + d + e \frac{d}{dt} P(t)$ avec $e > 0$.

Solution

On commence par poser les fonctions utiles.

```

> restart;
Qs:=P->a*P-b;#fonction d'offre
Qd:=(P,Pp)->-c*P+d+e*Pp;#fonction de demande
Ed:=(P,Pp)->m*(Qd(P,Pp)-Qs(P));#fonction d'excès de demande
      Qs:=P→aP−b
      Qd:=(P,Pp)→−cP+d+ePp
      Ed:=(P,Pp)→m(Qd(P,Pp)−Qs(P))

```

(6.1)

1) La résolution de l'équation différentielle retraçant l'évolution du prix ne pose aucun problème.

```

> eqd:=diff(P(t),t)=Ed(P(t),diff(P(t),t));#écriture de
l'équation d'évolution du prix
cint:=P(0)=P0;#condition initiale
dsolve({eqd,cint},P(t));#détermination de la trajectoire
solution générale
Psol:=subs(%,P(t));#récupération de l'expression de la
trajectoire

```

$$eqd := \frac{d}{dt} P(t) = m \left(-c P(t) + d + e \left(\frac{d}{dt} P(t) \right) - a P(t) + b \right)$$

$$cint := P(0) = P0$$

$$P(t) = e^{\frac{m(c+a)t}{-1+me}} \left(P0 - \frac{d+b}{c+a} \right) + \frac{d+b}{c+a}$$

$$Psol := e^{\frac{m(c+a)t}{-1+me}} \left(P0 - \frac{d+b}{c+a} \right) + \frac{d+b}{c+a}$$

(6.2)

2) Comme dans l'exercice précédent, le prix d'équilibre s'obtient par **solve** appliquée à l'équation $Ed(P, 0) = 0$ puisque la dérivée du prix est par définition nulle à l'équilibre dynamique.

```

> Pe:=solve(Ed(P,0)=0,P);

```

$$Pe := \frac{d+b}{c+a}$$

(6.3)

La convergence vers l'équilibre $Pe = \frac{d+b}{c+a}$ est assurée si l'exposant de l'exponentielle est négatif. Il

ya divergence dans le cas d'un exposant positif. On en déduit que si $e < \frac{1}{m}$, l'équilibre est globalement stable. Si $e > \frac{1}{m}$, l'équilibre est instable. Si $e = \frac{1}{m}$, alors l'équation différentielle **eqd** est dégénérée puisque la dérivée première de $P(t)$ disparaît; à toute date t , on a la trajectoire constante $P(t) = \frac{d+b}{c+a}$, qui convient si, dès le départ, $P(0) = \frac{d+b}{c+a}$. Maple confirme cette analyse du problème.

```

> assume(a>0,b>0,c>0,d>0,m>0,e<1/m);#écriture des hypothèses
sur les paramètres du modèle dans le cas convergent
limit(Psol,t=infinity);#recherche de la limite
      d~+b~
      c~+a~

```

(6.4)

```
> assume(a>0,b>0,c>0,d>0,m>0,e>1/m);#écriture des hypothèses
sur les paramètres du modèle dans le cas divergent
limit(Psol,t=infinity);#recherche de la limite
      
$$\text{signum}(P_0 c + P_0 a - d - b) \infty$$
 (6.5)
```

Cette dernière réponse est particulièrement intéressante dans la mesure où elle n'est pas totalement correcte. Si le prix initial est supérieur (respectivement : inférieur) au prix d'équilibre, il est certain que le prix devient infiniment grand (respectivement : infiniment petit). En revanche, si le prix de départ est aussi le prix d'équilibre, la trajectoire est monotone constante et converge bien vers l'équilibre.


Il reste à visualiser ces deux situations.

```
> a,b,c,d,e,m:=0.1,-5,0.25,10,5,0.1;#attribution de valeurs aux
paramètres dans le cas de la convergence
Psol;
Psoluce:=unapply(Psol,(P0,t));#Une trajectoire solution
dépend du prix initial et de la date
plot([seq(Psoluce((d+b)/(c+a)+i,t),i=-10..10,2.5)],t=0..100,
P=0..30);#représentation graphique
      
$$a, b, c, d, e, m := 0.1, -5, 0.25, 10, 5, 0.1$$

      
$$e^{-0.07000000000t} (P_0 - 14.28571429) + 14.28571429$$

      
$$Psoluce := (P_0, t) \rightarrow e^{-0.07000000000t} (P_0 - 14.28571429) + 14.28571429$$

```




```
> a,b,c,d,e,m:=0.1,-5,0.25,10,15,0.1;#attribution de valeurs
aux paramètres dans le cas de la divergence
Psol;
Psoluce:=unapply(Psol,(P0,t));
plot([seq(Psoluce((d+b)/(c+a)+i,t),i=-10..10,2.5)],t=0..25,P=
0..30);
```

$$a, b, c, d, e, m := 0.1, -5, 0.25, 10, 15, 0.1$$

$$e^{0.07000000000t} (P0 - 14.28571429) + 14.28571429$$

$$Psoluce := (P0, t) \rightarrow e^{0.07000000000t} (P0 - 14.28571429) + 14.28571429$$



Exercice E3 (démographie)

Enoncé

Le modèle démographique de Verhulst repose sur l'idée simple qu'en l'absence de mouvements migratoires une population varie selon son taux d'accroissement naturel, défini comme la différence entre le taux de natalité et le taux de mortalité. Le taux de natalité $n(t)$, rapport entre le nombre de naissances et la population totale $P(t)$, évolue suivant la règle $n(t) = a - b P(t)$ avec a et b positifs : plus la taille de la population est élevée, plus le taux de natalité est faible. Le taux de mortalité $m(t)$, rapport entre le nombre de décès et la population totale $P(t)$, évolue suivant la règle $m(t) = c + d P(t)$ avec c et d positifs : plus la taille de la population est élevée, plus le taux de mortalité est élevé.

- 1) Montrer que l'évolution de la population obéit à une loi de Bernouilli. Donner alors l'expression de la trajectoire pour la condition initiale $P(0) = P_0$.
- 2) Montrer qu'il existe deux états stationnaires pour la dynamique démographique.
- 3) Montrer que la convergence vers un de ces deux états dépend des paramètres a et c .

Solution

On commence par poser les fonctions de taux de natalité et de taux de mortalité.

```
> restart;
n:=P->a-b*P;#taux de natalité
```

$$\begin{aligned}
 m &:= P \rightarrow c + d * P; \# \text{taux de mortalité} \\
 n &:= P \rightarrow a - b * P \\
 m &:= P \rightarrow c + d * P
 \end{aligned}
 \tag{7.1}$$

1) L'évolution instantanée de la population est égale au taux d'accroissement naturel appliqué à la population de sorte que l'évolution démographique est donnée par l'équation différentielle $P'(t) = (n(t) - m(t))P(t)$.

En calant `infolevel[dsolve]` à la valeur 2, on voit que Maple reconnaît une équation différentielle de Bernoulli dont la résolution est immédiate.

```

> eqd:=diff(P(t),t)=P(t)*(n(P(t))-m(P(t)));#loi d'évolution de
la population
cin:=P(0)=P0;#condition initiale
infolevel[dsolve]:=2;#demande de détails sur la résolution de
l'EDO
dsolve({eqd,cin},P(t));#résolution de l'EDO
Pop:=subs(%,P(t));#expression de la trajectoire-solution

```

$$eqd := \frac{d}{dt} P(t) = P(t) (a - b P(t) - c - d P(t))$$

$$cin := P(0) = P0$$

$$infolevel_{dsolve} := 2$$

Methods for first order ODEs:

```

--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
<- Bernoulli successful

```

$$\begin{aligned}
 P(t) &= - \frac{P0 (a - c)}{-b P0 - d P0 - e^{-(a-c)t} a + e^{-(a-c)t} c + e^{-(a-c)t} b P0 + e^{-(a-c)t} d P0} \\
 Pop &:= - \frac{P0 (a - c)}{-b P0 - d P0 - e^{-(a-c)t} a + e^{-(a-c)t} c + e^{-(a-c)t} b P0 + e^{-(a-c)t} d P0}
 \end{aligned}
 \tag{7.2}$$

2) Cherchons les équilibres avec `solve`.

```

> eqs:=solve(rhs(eqd),P(t));

```

$$eqs := 0, \frac{a - c}{b + d} \tag{7.3}$$

Il y a deux équilibres de population. Le premier correspond à une population ... nulle, le second à un niveau $\frac{a - c}{b + d}$ qui est strictement positif si $a > c$.

3) Avec `limit`, on se convainc que la population converge vers l'état stationnaire positif si $a > b$ et converge vers 0, c'est à dire s'éteint, si $a \leq c$.

```

> limit(Pop,t=infinity) assuming a>0,b>0,c>0,d>0,a>c,P0>0;

```

$$\frac{a - c}{b + d} \tag{7.4}$$

```

> limit(Pop,t=infinity) assuming a>0,b>0,c>0,d>0,a<=c,P0>0;

```

$$\tag{7.5}$$

Exercice E4 (problème de Constantin)

Énoncé

Constantin Chilarescu est un économiste-mathématicien roumain spécialisé dans les modèles de croissance endogène. Il a souvent recours à Maple pour confirmer ses résultats théoriques, voire pour l'aider à résoudre des problèmes complexes. Le but de cet exercice est d'étudier une extension qu'il a donnée au modèle de Lucas-Uzawa et qui conduit à la forme réduite

$$\frac{d}{dt} z(t) = \frac{p z(t)}{b-a} - \frac{g z(t)^{2-b}}{(1-a)(b-a)} + \frac{d(1-a) z(t)^{1-a}}{b-a}$$
 où z est le capital par unité de travail efficace, a est l'élasticité du capital ($0 < a < 1$), b est l'élasticité du travail ($0 < b < 1$ et $a < b$), d est le taux de dépréciation du capital ($0 < d < 1$), p est le taux de préférence pour le présent ($p > 0$) et g est le taux de croissance du progrès technique ($g > 0$).

1) Mener une étude qualitative de la dynamique d'accumulation du capital en étudiant les propriétés

de la fonction $f(z) = \frac{p z}{b-a} - \frac{g z^{2-b}}{(1-a)(b-a)} + \frac{d(1-a) z^{1-a}}{b-a}$

2) On donne aux paramètres les valeurs suivantes : $a = 0.3$; $b = 0.35$; $d = 0.075$; $g = 1.05$; $p = 0.05$.

Mener une étude qualitative graphique de l'équation différentielle à l'aide de **dsolve** et de

DEplot.

Solution

1) Dans cette question, on considère la fonction définie par $f(z)$

$$= \frac{p z}{b-a} - \frac{g z^{2-b}}{(1-a)(b-a)} + \frac{d(1-a) z^{1-a}}{b-a}$$
 qui est définie et de classe C^∞ sur \mathbb{R}^+* . Son

tableau de variation permet de connaître le signe de $\frac{d}{dt} z(t)$.

> restart;

assume(z>0,a>0,a<1,b>0,b<1,a<b,d>0,g>0,p>0);#hypothèses sur les paramètres du modèle

f:=z->(p/(b-a))*z-(g/((1-a)*(b-a)))*z^(2-b)+d*(1-a)/(b-a)*z^(1-a);

fp:=diff(f(z),z);#calcul de la dérivée première

fs:=diff(f(z),z,z);#calcul de la dérivée seconde

$$f := z \rightarrow \frac{p z}{b-a} - \frac{g z^{2-b}}{(1-a)(b-a)} + \frac{d(1-a) z^{1-a}}{b-a}$$

$$fp := \frac{p}{b-a} - \frac{g z^{2-b} (2-b)}{(1-a)(b-a) z} + \frac{d(1-a)^2 z^{1-a}}{(b-a) z}$$

$$fs := -\frac{g z^{2-b} (2-b)^2}{(1-a)(b-a) z^2} + \frac{g z^{2-b} (2-b)}{(1-a)(b-a) z^2} + \frac{d(1-a)^3 z^{1-a}}{(b-a) z^2} - \frac{d(1-a)^2 z^{1-a}}{(b-a) z^2} \quad (8.1)$$

Commençons par étudier la concavité de f . L'étude directe du signe de la dérivée seconde ne donne pas grand chose (déception!).

> is(fs<0);

FAIL

(8.2)

Mais il est facile de voir que cette dérivée seconde est strictement négative car on a : $f''(z) = \frac{g z^{2-b} (2-b) (-1+b)}{(1-a)(b-a) z^2} - \frac{a d z^{1-a} (1-a)^2}{(b-a) z^2}$ avec $(-1+b) < 0$; $(2-b) > 0$; $g > 0$; $(1-a) > 0$; $(b-a) > 0$. Ainsi, pour des valeurs économiques plausibles des paramètres, la fonction f est strictement concave sur \mathbb{R}_+^* .

Cherchons à présent les particularités de la fonction dérivée première. En particulier, pour quelle(s) valeur(s) s'annule-t-elle?

La résolution directe de $fp = 0$ ne donne pas grand chose avec Maple (deuxième déception!) :

> solve(fp=0, z);

$$\text{RootOf}(-p z + p z a + 2 g z^{2-b} - g z^{2-b} b - d z^{1-a} + 3 d z^{1-a} a - 3 d z^{1-a} a^2 + d z^{1-a} a^3) \quad (8.3)$$

Posons alors $fp = -h(z) + j(z)$ avec $h(z) = \frac{p}{b-a} - \frac{g z^{1-b} (2-b)}{(1-a)(b-a)}$ et $j(z) = \frac{d (1-a)^2 z^{-a}}{(b-a)}$.

La fonction $-h(z)$ est strictement croissante de $-\frac{p}{b-a}$ à $+\infty$ et la fonction $j(z)$ est décroissante de $+\infty$ à 0. Alors, il existe une valeur unique de z , notée z^* , pour laquelle on a $fp = 0$. En conséquence, la fonction f passe par un maximum global unique. Elle est croissante de 0 à z^* , où elle passe par un maximum, puis décroît continûment. Comme

$$\lim_{z \rightarrow \infty} \left(\frac{p z}{b-a} - \frac{g z^{2-b}}{(1-a)(b-a)} + \frac{d (1-a) z^{1-a}}{b-a} \right) = \lim_{z \rightarrow \infty} z \left(\frac{p}{b-a} - \frac{g z^{1-b}}{(1-a)(b-a)} + \frac{d (1-a) z^{-a}}{b-a} \right) = +\infty \times -\infty = -\infty$$

, elle traverse l'axe des abscisses en un point unique qui est l'équilibre stationnaire de la dynamique en z . C'est un équilibre dynamique globalement stable sur \mathbb{R}_+^* puisque la dérivée de la fonction $z(t)$ est positive à gauche de la valeur stationnaire et elle est négative à droite.

2) On donne aux paramètres leurs valeurs par assignation séquentielle :

> a,b,d,g,p:=0.3,0.35,0.075,1.05,0.05;

$$a, b, d, g, p := 0.3, 0.35, 0.075, 1.05, 0.05 \quad (8.4)$$


Profitons-en pour confirmer graphiquement l'étude menée dans la question précédente. On récupère l'expression de $f(z)$:

> f(z);

$$1.000000000 z - 30.00000000 z^{1.65} + 1.050000000 z^{0.7} \quad (8.5)$$

Puis on demande la représentation graphique :

> plot(f(z), z=0..0.1);




Ici, l'état stationnaire est au point $z^* = 0.041$. On constate bien qu'il est globalement stable : quand $z^* \leq 0.041$, $z' \geq 0$ et quand $z^* > 0.041$, $z' < 0$.

Passons aux questions proprement dites. Les simulations numériques ne posent aucun problème particulier (ce qui atténue notre déception!). Avec **DEplot**, on peut tracer directement les champs directionnels avec des trajectoires où on choisit des valeurs initiales pour z . Le diagramme confirme bien que l'équilibre stationnaire est globalement stable.

```
> with(DEtools):
edo:=diff(z(t),t)=(p/(b-a))*z(t)-(g/((1-a)*(b-a)))*z(t)^(2-b)
+d*(1-a)/(b-a)*z(t)^(1-a);
DEplot(edo,z(t),t=0..2,[[0,0.1],[0,0.08],[0,0.06],[0,0.04],
[0,0.02]],stepsize=0.01,thickness=0,linecolor=black);
```

$$edo := \frac{d}{dt} z_{\sim}(t) = 1.000000000 z_{\sim}(t) - 30.00000000 z_{\sim}(t)^{1.65} + 1.050000000 z_{\sim}(t)^{0.7}$$




Une deuxième possibilité est offerte par **dsolve** avec l'option **numeric** à partir d'une condition initiale donnée (ici $z(0) = 0.08$).

```
> soluce:=dsolve({edo,z(0)=0.08},z(t),numeric);
                soluce:=proc(x_rkf45) ... end proc
```

(8.6)

```
> f:=x->subs(soluce(x),z(t));
graf:=plot(f,0..2,0..0.1):with(plots):display(graf);
                f:=x->subs(soluce(x),z(t))
```



Encore une fois, on voit qu'il y a convergence très rapide vers l'état stationnaire.